

Code

```
#include <Wire.h>

#include "RTClib.h"

RTC_DS1307 rtc;

// Segment pins A to G

const int segmentPins[7] = {2, 3, 4, 5, 6, 7, 8};

// Digit cathode control pins

const int digitPins[4] = {9, 10, 11, 12};

// Segment codes for 0-9

const byte digitCode[10] = {

    B00111111, // 0

    B00000110, // 1

    B01011011, // 2

    B01001111, // 3

    B01100110, // 4

    B01101101, // 5

    B01111101, // 6

    B00000111, // 7

    B01111111, // 8

    B01101111 // 9

};

void setup() {

    Wire.begin();

    rtc.begin();

    // rtc.adjust(DateTime(F(_DATE), F(_TIME))); // Uncomment once to set time

    for (int i = 0; i < 7; i++) pinMode(segmentPins[i], OUTPUT);

    for (int i = 0; i < 4; i++) pinMode(digitPins[i], OUTPUT);

}
```

```

void loop() {
    DateTime now = rtc.now();
    int hour = now.hour();
    int minute = now.minute();
    int digits[4] = {
        hour / 10,
        hour % 10,
        minute / 10,
        minute % 10
    };
    for (int i = 0; i < 4; i++) {
        showDigit(i, digits[i]);
        delay(5);
        clearDigits();
    }
}

void showDigit(int digitIndex, int number) {
    byte segments = digitCode[number];
    for (int s = 0; s < 7; s++) {
        digitalWrite(segmentPins[s], bitRead(segments, s));
    }
    digitalWrite(digitPins[digitIndex], LOW); // Enable digit
}

void clearDigits() {
    for (int i = 0; i < 4; i++) {
        digitalWrite(digitPins[i], HIGH); // Disable all
    }
}

```